Online cloud detection in solar full-disc white-light images

Miroslav Klvaňa and Michal Sobotka

Astronomical Institute of the Czech Academy of Sciences, Ondřejov

INTRODUCTION

Ground-based solar observations can be significantly affected by clouds. The shadows of clouds decrease the brightness of parts of the solar disc and sometimes obscure them completely. For automated routine synoptic observations, there is a need to detect the presence of cloud shadows and to evaluate how strong their effect is. For this purpose, it is useful to define four degrees of the image degradation by clouds (a cloud score):

0 – Clean Sun (Fig. 1a), no clouds; the images can be used for automatic solar activity detection.
1 – Weak clouds (Fig. 1b); the images cannot be used for automatic detection but they are still useful.
3 – Strong clouds (Fig. 1c, d); the images are severely affected and it is not worth to store them.
4 – No Sun; the solar disc is completely obscured by clouds.

Feng et al. (2014) published a method of automatic detection of cloud shadows and their removal in full-disc images, based on a comparison of observed images with a reference one. The method, which requires a centering of the full-disc image and an analysis of its radial profiles, is computationally expensive and it is intended mostly for a post-processing.

We present a method that analyses each observed image individually and does not need a reference frame or a cross-comparison between the images. This makes it possible to determine the cloud score immediately after taking the image and to decide about its further processing.

![Fig. 1](image)

Fig. 1 – Examples of the cloud effects: a - clean Sun; b - weak clouds; c, d - strong clouds.
DATA SETS

To elaborate the method, we used several data sets of white-light images obtained by the Ondřejov D = 50 mm, f = 500 mm full-disc patrol telescope with a wavelength band width of 7 nm centered at 536 nm. The 8-bit images are 1600 x 1200 pixels large and the image scale is 1.83”/pixel. Six data sets include periods of clean Sun, weak cirrus clouds, and other clouds with different transmissivity. Two long data sets with 1010 and 2270 frames cover periods with variable cloudiness.

CRITERIA

Minimum brightness – this criterion checks if the brightness signal is sufficient for further evaluation. Its value, maxdark, is set well above the dark noise, in the case of 8-bit data to 10. If the observed intensity maximum \( I_{\text{max}} \) is smaller than maxdark, the Sun is completely obscured by clouds and no further evaluation is made.

Criterion 1 – A ratio of average intensities of the solar disc and of the rest of frame.

\[
\text{crit}1 = \frac{\text{mean}(I_{\text{disc}})}{\text{mean}(I_{\text{non-disc}})} \quad (1)
\]

The pixels belonging to the solar disc are separated from the non-disc pixels by the threshold intensity 0.2 \( I_{\text{max}} \). This threshold is based on the theoretical center-to-limb variation curve.

The criterion, a contrast of the solar disc to its surroundings, is independent of image scale but it depends on number of bits per data pixel. Due to the scattered light, it may slightly depend on the field of view, i.e., on the number of non-disc pixels. In our data, the area covered by the disc is 43.7 % of the full FOV.

For a correct exposure time, when nearly the whole dynamical range of intensities is used, and for clean Sun conditions, the intensity ratio is 50–51 in our 8-bit data.

The typical values of Criterion 1 in 8-bit data are as follows:

\( \text{crit}1 = 0 \) ... is set for too dark images, \( I_{\text{max}} < \text{maxdark} \), no Sun

\( \text{crit}1 = 20 \) ... is the minimum value for using Criteria 2 and 3; \( 0 < \text{crit}1 < 20 \) means strong clouds

\( \text{crit}1 = 40 \) ... is the minimum value for a clean Sun (it depends on exposure time and Sun elevation)

\( \text{crit}1 = 50 \) ... is typical for a clean Sun

These values have to be determined experimentally.

Criterion 2 – A ratio of observed and calculated disc areas. It is applied only when \( \text{crit}1 > 20 \).

\[
\text{crit}2 = \frac{N_{\text{pix,obs}}}{N_{\text{pix,calc}}} \quad \text{for } \text{crit}1 > 20
\]

\[
\text{crit}2 = 0 \quad \text{for } \text{crit}1 < 20.
\]

Here, \( N_{\text{pix,obs}} \) is a number of pixels brighter than 0.2 \( I_{\text{max}} \) (disc pixels) and \( N_{\text{pix,calc}} = \pi (r \odot / \text{scale})^2 \) is the calculated disc area depending on the image scale and the apparent angular solar radius \( r \odot \), calculated by an ephemeris program for the date of observation.

When \( \text{crit}2 < 1 \), a part of the solar disc is covered by a strong cloud (Fig. 1d). If \( \text{crit}2 = 1 \pm 0.01 \), the full disc is seen and the necessary condition for clean Sun or weak clouds is met.

Criterion 3 – A normalized standard deviation of the full-disc intensity. Applied only when \( \text{crit}2 \approx 1 \).

First, the full-disc image is smoothed by a boxcar of 0.5’x 0.5’ to remove random noise and diminish the effect of sunspot groups and plages. Then the standard deviation of disc intensities is calculated and divided by their mean value.

\[
\text{crit}3 = \frac{\text{stdev}(I_{\text{disc,smoothed}})}{\text{mean}(I_{\text{disc,smoothed}})} \quad (3)
\]

The value of \( \text{crit}3 \) is given by the center-to-limb variation and its perturbation by clouds. The unperturbed value is close to 0.205 (clean Sun) and for weak clouds it ranges between the limits
0.18–0.22. Values out of this range correspond to strong clouds (Fig. 1c). This criterion is used to distinguish between the clean-Sun conditions (in the range 0.199–0.202), weak clouds (in the range 0.18–0.22), and strong clouds. These values have to be determined experimentally.

**CLOUD SCORE**

The cloud score is obtained by a combination of conditions derived from the criteria described above:

- **Condition for minimum brightness** \( cr0 = 1 \) for \( I_{max} > maxdark \) else \( cr0 = 0 \)
- **Condition for clean Sun** \( cr1 = 1 \) for \( crit1 > 40 \) else \( cr1 = 0 \)
- **Condition for clean Sun or weak clouds** \( cr2 = 1 \) for \( crit2 = 1 \) else \( cr2 = 0 \)
- **Condition for clean Sun or weak clouds** \( cr31 = 1 \) for \( crit3 = 0.205 \) else \( cr31 = 0 \)
- **Condition for weak clouds** \( cr32 = 1 \) for \( 0.18 < crit3 < 0.22 \) else \( cr32 = 0 \)

The combination of conditions corresponding to individual cloud scores is shown in Table 1.

<table>
<thead>
<tr>
<th>Cloud score</th>
<th>( cr0 )</th>
<th>( cr1 )</th>
<th>( cr2 )</th>
<th>( cr31 )</th>
<th>( cr32 )</th>
</tr>
</thead>
<tbody>
<tr>
<td>0 – clean Sun</td>
<td>1</td>
<td>1</td>
<td>1</td>
<td>1</td>
<td>1</td>
</tr>
<tr>
<td>1 – weak clouds</td>
<td>1</td>
<td>0</td>
<td>1</td>
<td>1</td>
<td>1</td>
</tr>
<tr>
<td>1 – weak clouds</td>
<td>1</td>
<td>0 or 1</td>
<td>1</td>
<td>0</td>
<td>1</td>
</tr>
<tr>
<td>2 – strong clouds</td>
<td>1</td>
<td>0 or 1</td>
<td>1</td>
<td>0</td>
<td>0</td>
</tr>
<tr>
<td>2 – strong clouds</td>
<td>1</td>
<td>0 or 1</td>
<td>0</td>
<td>n/a</td>
<td>n/a</td>
</tr>
<tr>
<td>3 – no Sun</td>
<td>0</td>
<td>n/a</td>
<td>n/a</td>
<td>n/a</td>
<td>n/a</td>
</tr>
</tbody>
</table>

**RESULTS**

The criteria, conditions, and their relations were included in the CLOUD_TEST routine written in the IDL language (see Appendix for the source code). The code, running on an ordinary PC, analyzes 35 frames per second when the frames are stored on a local disc and 15 fps if they are transferred by a local network. This is sufficient for online processing. Three data sets, 130528 (161 frames), 130608 (1010 frames), and 130613 (2273 frames), covering periods of clean Sun, weak clouds, strong clouds, and the totally obscured Sun, were used to check the performance of the method. Plots of \( crit1 \), \( crit2 \), and \( crit3 \), together with the cloud score, are shown in Fig. 2 for the three data sets.

The results were compared with the visual assessment of processed images. The errors occur mainly at subtle transitions between the clean Sun and weak clouds and between the weak and strong clouds. The decisive criterion is \( crit3 \), which is the weakest of the criteria due to the fact that the clouds may both increase and decrease the normalized standard deviation of disc intensities. These changes may compensate each other in some cases. In the test runs, 3.3% of weak-cloud frames were classified erroneously as a clean Sun and 2.7% of strong-cloud images as weak clouds. These plausible results show that the suggested method might be useful.
Fig. 2 – Plots of criteria and the cloud score in three data sets. Crit1 (dashed) was set to 43 for the sets 130528, 13608 and to 37 for 130613. Crit2 (long dash) is multiplied by factor of 20 and Crit3 (solid line) by factor of 100. Small squares at the bottom of the plots denote the cloud score 0–3.

References

APPENDIX
IDL program CLOUD_TEST.PRO

PRO cloud_test, directory, cloud, mov
;; Test of clouds in full-disc frames in white light
;; INPUTS: directory - frame folder name (string)
;; OUTPUT: cloud - cloud score: 0-clean Sun, 1-weak cloud, 2-strong cloud, 3-no Sun
;; mov - movie (byte) of downscaled images with cloud score and Crit.3
;; METHOD:
;; 1. Ratio of average intensities on the disc and out of the disc (Criterion 1)
;; 2. Area above an intensity threshold 0.2 / calculated full-disc area (Criterion 2)
;; 3. Normalized standard deviation of disc intensity (Criterion 3)
;; CALLED: L0B0P0.pro to calculate theoretical solar radius for a given date.
; 18 August 2016, Michal
; PARAMETERS -------------------------------
image_scale=1.8303877 ; arcsec/pixel
maxdark=10 ; "no-Sun" threshold to avoid division by zero (counts)
minbright=20 ; minimum average I_disc / average I_nodisc

; evaluation criteria limits
cri1min=43 ; 37-43, minimum average I_disc / average I_nodisc for a clean Sun
; - CAUTION: sensitive to exposure changes and the Sun elevation
cri2min=0.99 ; Criterion 2 should be close to 1 for full-disc visibility
cri2max=1.01 ; (more or less fixed parameters)
cri3min1=0.199 ; Criterion 3 is slightly above 0.20 for clean Sun
cri3max1=0.202 ; (to be tuned but quite stable)
cri3min2=0.18 ; Criterion 3 limits for weak clouds
cri3max2=0.22 ; (to be tuned)

; INPUT BLOCK -----------------------------
path='TBD'
list=FILE_SEARCH(path+directory+'*.bmp',count=cnt) ; list of files

; here the date is derived from the directory name:
year =2000+fix(strmid(directory,0,2))
month=fix(strmid(directory,2,2))
day=fix(strmid(directory,4,2))
hour=12 ; fixed

; CALCULATED DISC AREA -------------------
rdisk_pixels=rdisk_arcsec/image_scale
smooth_box=rdisk_pixels/32. ; smoothing parameter approx. 30"

; OUTPUT ARRAYS --------------------------
crit1=fltarr(cnt)
crit2=fltarr(cnt)
crit3=fltarr(cnt)
cloud=intarr(cnt)
mov=bytarr(400,300,cnt) ; check-movie array
window,0,xs=400,ys=300

; GENERAL LOOP ON FRAMES ----------------
for i=0,cnt-1 do begin
ima=READ_BMP(list(i))
ima=fix(ima) ; conversion to integer
imax=max(ima) ; maximum intensity
limit=0.2*imax ; disc threshold (a more or less fixed parameter)

wdisk=where(ima gt limit,cdisk) ; cdisk = number of Npix(I>0.2) disc pixels
wout=where(ima lt limit,cout) ; cout = number of out-of-disc pixels

cri1=0 ; condition variables predefined as false
cri2=0
cri31=0
crit32=0

; Criterion 1 - ratio avg. disc intensity / avg. non-disc intensity
if imax gt maxdark then $
crit1(i)=(total(ima(wdisk))/cdisk)/(total(ima(wout))/cout)\$
else crit1(i)=0. ; imax < maxdark - no Sun
if crit1(i) gt cri1min then cri1=1 ; Criterion 1 in clean Sun range
if crit1(i) gt minbright then begin ;the Sun is not too much covered by clouds
    ;Criterion 2 - ratio Npix(I>0.2) / Npix(disc) to detect a partial cover of the disc
    crit2(i)=float(cdisk)/ndisk
    if (crit2(i) gt cri2min) and (crit2(i) lt cri2max) then cr2=1 ;full disc seen
endif
if cr2 then begin ;full disc visible
    ;Criterion 3 - normalized standard deviation of disc intensity to detect clouds
    sima=smooth(ima,smooth_box) ;smoothing of the frame to remove noise
    mom=moment(sima(wdisk),maxmoment=2) ;calculation of mean mom(0) and variance mom(1)
    crit3(i)=sqrt(mom(1))/mom(0) ;stddev = sqrt(variance)
    if (crit3(i) gt cri3min1) and (crit3(i) lt cri3max1) then cr31=1
    ;Criterion 3 in clean Sun range
    if (crit3(i) gt cri3min2) and (crit3(i) lt cri3max2) then cr32=1
    endif
; EVALUATION ------------------------
; cloud = 0 - clean Sun
; cloud = 1 - weak clouds
; cloud = 2 - strong clouds
; cloud = 3 - no Sun
if crit1(i) eq 0. then cloud(i)=3 else begin ;no Sun
    case 1 of
    (cr1 and cr2 and cr31)        : cloud(i)=0 ;clean Sun
    (cr2 and cr32 and (not cr31)) : cloud(i)=1 ;weak clouds
    (cr2 and cr31 and (not cr1))  : cloud(i)=1
    abs(not cr2)                  : cloud(i)=2 ;strong clouds
    (cr2 and (not cr32))          : cloud(i)=2
    else                          : cloud(i)=10 ;!undefined!
    endcase
endelse
; A MOVIE TO CHECK ---------------------
ima=congrid(ima,400,300)
tvscl,ima
xyouts,20,20,strtrim(cloud(i),2)+'   '+strtrim(crit3(i),2),charsiz=1.6,/device
a=tvrd()
mov(*,*,i)=a

; OUTPUT PLOTS ------------------------
window,0,xs=1000,ys=700
plot,crit1,line=2,/xst,col=0,back=255,charsiz=1.6,title=directory, $
    xtitle='Frames',ytitle='Crit1, Crit2*20, Crit3*100, Cloud'
oplot,crit2*20,col=0,line=5
oplot,crit3*100,col=0
oplot,crit1,psym=1,psymx=0.6,col=0
oplot,cloud,col=0,psym=6,psymx=0.6
END